
6.001: Structure and Interpretation of 
Computer Programs

• Today
– Building a new language using data and 

procedure abstractions

Themes to be integrated
• Data abstraction

– Separate use of data structure from details of data 
structure

• Procedural abstraction
– Capture common patterns of behavior and treat as black 

box for generating new patterns
• Means of combination

– Create complex combinations, then treat as primitives 
to support new combinations

• Use modularity of componentsto create new language for 
particular problem domain

10/5/2004 6.001 SICP 3/41

Our target – the art of M. C. Escher

My buddy George

A procedural definition of George
(define (george rect)

(draw-line rect .25 0 .35 .5)
(draw-line rect .35 .5 .3 .6)
(draw-line rect .3 .6 .15 .4)
(draw-line rect .15 .4 0 .65)
(draw-line rect .4 0 .5 .3)
(draw-line rect .5 .3 .6 0)
(draw-line rect .75 0 .6 .45)
(draw-line rect .6 .45 1 .15)
(draw-line rect 1 .35 .75 .65)
(draw-line rect .75 .65 .6 .65) 
(draw-line rect .6 .65 .65 .85)
(draw-line rect .65 .85 .6 1)
(draw-line rect .4 1 .35 .85)
(draw-line rect .35 .85 .4 .65)
(draw-line rect .4 .65 .3 .65)
(draw-line rect .3 .65 .15 .6)
(draw-line rect .15 .6 0 .85))

Yuck!!

Data abstractions for lines

(2, 3)

(define p1 (make-vect 2 3))

(xcor p1) àààà 2

(ycor p1) àààà 3

(define s1 (make-segment p1 p2))

(xcor (start-segment s1)) àààà 2

(ycor (end-segment s1)) àààà 4

(5, 4)

(define p2 (make-vect 5 4))

A better George(define p1 (make-vect .25 0))

(define p2 (make-vect .35 .5))

(define p3 (make-vect .3 .6))

(define p4 (make-vect .15 .4))

(define p5 (make-vect 0 .65))

(define p6 (make-vect .4 0))

(define p7 (make-vect .5 .3))

(define p8 (make-vect .6 0))

(define p9 (make-vect .75 0))

(define p10 (make-vect .6 .45))

(define p11 (make-vect 1 .15))

(define p12 (make-vect 1 .35))

(define p13 (make-vect .75 .65))

(define p14 (make-vect .6 .65))

(define p15 (make-vect .65 .85))

(define p16 (make-vect .6 1))

(define p17 (make-vect .4 1))

(define p18 (make-vect .35 .85))

(define p19 (make-vect .4 .65))

(define p20 (make-vect .3 .65))

(define p21 (make-vect .15 .6))

(define p22 (make-vect 0 .85))

(define george-lines

(list (make-segment p1 p2)
(make-segment p2 p3)
(make-segment p3 p4)
(make-segment p4 p5)
(make-segment p6 p7)
(make-segment p7 p8)
(make-segment p9 p10)
(make-segment p10 p11)
(make-segment p12 p13)
(make-segment p13 p14)
(make-segment p14 p15)
(make-segment p15 p16)
(make-segment p17 p18)
(make-segment p18 p19)
(make-segment p19 p20)
(make-segment p20 p21)
(make-segment p21 p22)))

• Have isolated elements of abstraction

•Could change a point without having to 
redefine segments that use it

•Have separated data abstraction from its use

Gluing things together

For pairs, use a cons:

For larger structures, use a list:

1 2 3 4

(list 1 2 3 4)

(cons 1 (cons 2 (cons 3 (cons 4 nil))))
10/5/2004 6.001 SICP 10/41

Properties of data structures
• Contract between constructor and selectors
• Property of closure:

– A list is a sequence of pairs, ending in the 
empty list, nil.

– Consing anything onto a list results in a list (by 
definition)

– Taking the cdr of a list results in a list (except 
perhaps for the empty list)

Completing our abstraction

Points or vectors:

(define make-vect cons)

(define xcor car)

(define ycor cdr)

Line segments:

(define make-segment list)

(define start-segment car)

(define end-segment cadr)

Drawing in a rectangle or frame

(1, 1)

(0, 0)

origin

y axis

x axis

Generating the abstraction of a frame
Rectangle:

(define make-rectangle list)

(define origin car)

(define horiz cadr)

(define vert caddr)

Picture:

(define some-primitive-picture

(lambda (rect)

<draw some stuff in rect >

))

What happens if we change an abstraction?

(define make-vect list)

(define xcor car)

(define ycor cadr)

Note that this still 
satisfies the contract

What else needs to change in our system?BUPKIS, 
NADA, 
NOTHING

What is a picture?

• Could just create a general procedure to draw 
collections of line segments

• But want to have flexibility of using any frame to 
draw in frame

• SO– we make a picture be a procedure!!
• Captures the procedural abstraction of drawing 

data within a frame

Manipulating vectors

+vect
scale-vect

Reassemble new parts

Compute more primitive 
operation

(define (+vect v1 v2)

(make-vect (+ (xcor v1) (xcor v2))

(+ (ycor v1) (ycor v2))))

(define (scale-vect vect factor)

(make-vect (* factor (xcor vect))

(* factor (ycor vect))))

(define (-vect v1 v2)

(+vect v1 (scale-vect v2 –1)))

(define (rotate-vect v angle)

(let ((c (cos angle))

(s (sin angle)))

(make-vect (- (* c (xcor v))

(* s (ycor v)))

(+ (* c (ycor v))

(* s (xcor v))))))

Select parts

Creating a picture

segments

make-picture

Picture procrect

picture on screen

The picture abstraction
(define (make-picture seglist)

(lambda (rect)

(for-each

(lambda (segment)

(let ((b (start-segment segment))

(e (end-segment segment)))

(draw-line rect

(xcor b)

(ycor b)

(xcor e)

(ycor e))))

seglist)))

Higher order 
procedure

For-each is like map, 
except it doesn’t collect a 
list of results, but simply 
applies procedure to each 
element of list for effect

Drawing lines is just algebra

• Drawing a line is just some algebra.  If a rectangle has an 
origin o, a horizontal axis u and a vertical axis v then a 
point p, with components x and y gets mapped to the point: 
o + xu + yv

(1, 1)

(0, 0)

origin

y axis

x axis

A better George
Remember we have george-lines from before

So here is George!

(define g (make-picture george-lines))

Operations on pictures

rotate

O
H

V

O’

H’

V’

Operations on pictures

(define (rotate90 pict)

(lambda (rect)

(pict (make-rectangle

(+vect (origin rect)

(horiz rect))

(vert rect)

(scale-vect (horiz rect) –1))))

(define (together pict1 pict2)
(lambda (rect) 

(pict1 rect)
(pict2 rect)))

A Georgian mess!

(draw (together 

g

(rotate90 g)))

Operations on pictures

PictA:

PictB:

beside

above

Creating a picture

Picture proc

Picture proc

beside

picture on screen

rect

10/5/2004 6.001 SICP 27/41

More procedures to combine pictures:

(define (beside pict1 pict2 a)

(lambda (rect)

(pict1 

(make-rectangle

(origin rect)

(scale-vect (horiz rect) a)

(vert rect)))

(pict2

(make-rectangle

(+vect

(origin rect)

(scale-vect (horiz rect) a))

(scale-vect (horiz rect) (- 1 a))

(vert rect)))))

(define (above pict1 pict2 a)

(rotate270

(beside (rotate90 pict1)

(rotate90 pict2)

a)))

Pictures have a 
closure property!

Big brother

(define big-bro

(beside g

(above empty-picture g .5)

.5))

A left-right flip

H

V flip
V’

H’

(define (flip pict)

(lambda (rect)

(pict (make-rectangle

(+vect (origin rect) (horiz rect))

(scale-vect (horiz rect) –1)

(vert rect)))))

(define acrobats

(beside g

(rotate180 (flip g))

.5))

(define 4bats

(above acrobats

(flip acrobats)

.5))

Recursive combinations of pictures

(define (up-push pict n)

(if (= n 0)

pict

(above (up-push pict (- n 1))

pict

.25)))

Pushing George around

Pushing George around

(define (right-push pict n)

(if (= n 0)

pict

(beside pict

(right-push pict (- n 1))

.75)))

Pushing George into the corner
(define (corner-push pict n)

(if (= n 0)

pict

(above 

(beside

(up-push pict n)

(corner-push pict (- n 1))

.75)

(beside

pict

(right-push pict (- n 1))

.75)

.25)))

Pushing George into a corner

(corner-push 4bats 2)

Putting copies together
(define (4pict p1 r1 p2 r2 p3 r3 p4 r4)

(beside

(above

((repeated rotate90 r1) p1)

((repeated rotate90 r2) p2)

.5)

(above

((repeated rotate90 r3) p3)

((repeated rotate90 r4) p4)

.5)

.5))

(define (4same p r1 r2 r3 r4)

(4pict p r1 p r2 p r3 p r4))
(4same g 0 1 2 3)

(define (square-limit pict n)
(4same (corner-push pict n)

1 2 0 3))

(square-limit 4bats 2)

“Escher” is an embedded language

(define …)

(lambda …)

(define …)

(lambda …)

(define …)

(lambda …)

Abstraction
Naming

Creation

Together, beside, 
…,
And Scheme 
mechanisms

Cons, car, cdr(p a b)Combinations

Rotate90, …+, map, …Primitive 
procedures

Half-line, 
George, other 
pictures

Nil3, #f, georgePrimitive data

Picture languageScheme dataScheme


